
8 オイラー・ラグランジュ方程式

オイラー・ラグランジュ方程式の一般形を，次に示す．

d

dt

(
∂L
∂q̇i

)
− ∂L

∂qi
+

∂D
∂q̇i

= Fi (i = 1, · · · , n) (8.1)

q = [qi]はこのあとすぐ述べる一般化座標である．その他の諸量を，表 1にまとめておく．
これらの具体形を用意し，(8.1)に代入すると運動方程式が求まる．一般化力F = [Fi]に
ついては 9節の段階で述べる．

表 1: オイラー・ラグランジュ方程式の諸量
ラグランジュ関数 L L = T − U (T：運動エネルギー, U：ポテンシャル)

並進 1
2m|ẋ|2 x, θ,ω, I, J

運動エネルギー T 回転 (2次元) 1
2Iθ̇

2 I は慣性モーメント は慣性系で

回転 (3次元) 1
2ω · (Jω) ω は角速度ベクトル

J は慣性テンソル
測った成分

重力 mg h

ポテンシャル U 線形ばね 1
2k x

2

クーロン摩擦 1
2νR sgn(ẋ)ẋ

散逸関数D 粘性抵抗 1
2c ẋ

2

慣性抵抗 1
2D |ẋ|ẋ2

一般化力F = [Fi]
∑
k

∂xk
∂qi

fk 9節 p23

8.1 一般化座標 q = [qi]

例えば，平面上の剛体の姿勢は，位置ベクトル x = [xi]と回転角 θで完全に定まる．こ
の剛体の機構学的な自由度は 3である．そこで，この 3つの変数を並べた 3次元ベクトル，

q =

q1q2
q3

 :=

x1 [m]

x2 [m]

θ [rad]

 とか

 θ [rad]

x1 [m]

x2 [m]


を導入する．通常の空間座標とは異なり，単位の違う量が同列に並んでいる．このように，
機構学的な自由度を表す変数を，単位を気にしないで，ただ並べて作った数ベクトル qを，
一般化座標 (generalized coordinate)という．
変数のとり方や並べ順は，各人で好きに決めてよい．例えば，ばねで継がれた 3質点の
水平運動を考えるとき，図 1のような 2種類のとり方がありうる 1)．左は原点Oからの絶
対距離を並べた一般化座標，右は相対距離を並べた一般化座標である．いずれも 3質点の
機構学的な自由度を過不足なく表現できているので，どちらを採用してもよい．運動方程
式の解として，絶対距離が欲しい人は左を，相対距離が欲しい人は右を採用するだろう．
もちろん，物理的には同じ運動なのだから，運動方程式を解いたあとに相互に変換する
ことも可能である．解析力学の強みは，こうした二度手間を省き，各人のニーズに合せて，
欲しい変量を解として持つ運動方程式を導けるところにある．

1)もちろん他の取り方もある．考えてみよ！

(a) 絶対距離方式 (慣性系) (b) 相対距離方式 (非慣性系)

図 1: 一般化座標の例

8.2 座標変換 — 慣性座標と一般化座標

ニュートンの第 2法則mẍ = f が成立する座標系を慣性系と呼んだ．これで測った座
標を，慣性座標と呼ぼう．図 1(b)の原点Oを慣性系の原点とすると，q1は慣性座標だが，
q2, q3は慣性座標ではない．これを忘れると，運動エネルギー T の計算を間違う．
なぜなら，表 1の諸量のうち，運動エネルギー T の定義式は，慣性座標で書かれたもの
だからだ．ゆえに，例えば，図 1(b)の中央の質点の運動エネルギーは，

(誤) T =
m

2
q̇22

(正) T =
m

2
ẋ22 =

m

2
(q̇1 + q̇2)

2

のように，慣性座標 x2から計算しなければならない．
この種の見落しを防ぐためにも，一般化座標q = [qi]をとったら，必ず，慣性座標x = [xi]

への座標変換を書き下しておく．図 1については，x1x2
x3

 =

q1q2
q3

 ※図 1 (a)，恒等変換

x1x2
x3

 =

 q1

q1 + q2

q1 + q2 + q3

 ※図 1 (b)

となる．(a)の場合は，慣性座標 [xi]をそのまま一般化座標 [qi]としているが，もちろん
こうした取り方でもよい．
それ以外の，U，D，Fi の定義式は，慣性座標でなくとも使える．ようするに，定義式
がニュートンの第 2法則から導出されたものは，いったん慣性座標に戻して計算する．

例題 3.1 2節リンクの一般化座標 q = [qi]を，次のように 2種類とる．左は空間に対する
絶対角によるもの，右は相対角によるものである．

(a) 絶対角方式 (b) 相対角方式

19



中間節の慣性座標を x1 = (x11, x12)
T，先端の慣性座標を x2 = (x21, x22)

T とする． qか
ら x1,x2への座標変換を求めよ．

▶解答例 絶対角方式の一般化座標に対して，
x11

x12

x21

x22

 =


l1 sin q1
l1 cos q1

l1 sin q1 + l2 sin q2
l1 cos q1 + l2 cos q2

 ※ (a) 絶対角方式 (8.2)

相対角方式の一般化座標に対して，
x11

x12

x21

x22

 =


l1 sin q1
l1 cos q1

l1 sin q1 + l2 sin(q1 + q2)
l1 cos q1 + l2 cos(q1 + q2)

 ※ (b) 相対角方式 (8.3)

となる．

8.3 車輪型倒立ロボット

図の座標系を慣性系とする．一般化座標を (α, β)とすると，各質点の直交座標は，

xM =

[
x

r

]
, xm = Trans(xM )Rot(−β)

[
0

l

]
=

[
x+ l sinβ

r + l cosβ

]
, x = rα (8.4)

となる．

例題 3.2 オイラー・ラグランジュ方程式 (8.1) p18を利用して運動方程式を導け．

▶解答例 まず，(8.4)を時間微分すると，速度は，

ẋM =

[
ẋ
0

]
, ẋm =

[
ẋ+ lβ̇ cosβ

−lβ̇ sinβ

]
, ẋ = rα̇ (8.5)

となり，運動エネルギーは，

T =
M

2
ẋ2
M +

m

2
ẋ2
m +

I

2
α̇2 (8.6)

=
M

2
ẋ2 +

m

2

(
(ẋ+ lβ̇ cosβ)2 + (−lβ̇ sinβ)2

)
+

I

2
α̇2 (8.7)

=
M

2
ẋ2 +

m

2

(
ẋ2 + 2lẋβ̇ cosβ + l2β̇2

)
+

I

2
α̇2 (8.8)

=
Mr2

2
α̇2 +

m

2

(
r2α̇2 + 2lrα̇β̇ cosβ + l2β̇2

)
+

I

2
α̇2 (8.9)

=
(M +m)r2 + I

2
α̇2 +mlrα̇β̇ cosβ +

ml2

2
β̇2 (8.10)

となる．ポテンシャルは，

U = mgl cosβ + C (C は定数), (8.11)

となる．これより，この系のラグランジュ関数は，

L = T − U =
(M +m)r2 + I

2
α̇2 +mlrα̇β̇ cosβ +

ml2

2
β̇2 −mgl cosβ − C (8.12)

となる．これをオイラー・ラグランジュ方程式 (8.1) p18に代入すると運動方程式が求まる．
実際，必要な偏微分を計算すると，

∂L
∂α̇

= {(M +m)r2 + I}α̇+mlrβ̇ cosβ

=⇒ d

dt

∂L
∂α̇

= {(M +m)r2 + I}α̈+mlr(β̈ cosβ − β̇2 sinβ),
∂L
∂α

= 0,

∂L
∂β̇

= mlrα̇ cosβ +ml2β̇ =⇒ d

dt

∂L
∂β̇

= mlr(α̈ cosβ − α̇β̇ sinβ) +ml2β̈,

∂L
∂β

= −mlrα̇β̇ sinβ +mgl sinβ

となる．また，関節トルク T が自由度 αと β に発生する一般化力をそれぞれ Fα,Fβ と表記
しておく．これらを (8.1)に代入すると運動方程式は，

d

dt

∂L
∂α̇

− ∂L
∂α

={(M +m)r2 + I}α̈+mlr(β̈ cosβ − β̇2 sinβ)

= {(M +m)r2 + I}α̈+ (mlr cosβ)β̈ −mlrβ̇2 sinβ = Fα (8.13)

d

dt

∂L
∂β̇

− ∂L
∂β

=mlr(α̈ cosβ − α̇β̇ sinβ) +ml2β̈ − (−mlrα̇β̇ sinβ +mgl sinβ)

= (mlr cosβ)α̈+ml2β̈ −mgl sinβ = Fβ (8.14)

となる．ベクトル形式で加速度 α̈, β̈ をくくり出すと，運動方程式は，[
(M +m)r2 + I mlr cosβ

mlr cosβ ml2

]
︸ ︷︷ ︸

A

[
α̈

β̈

]
︸︷︷︸
q̈

=

[
mlrβ̇2 sinβ
mgl sinβ

]
︸ ︷︷ ︸

b

+

[
Fα

Fβ

]
︸ ︷︷ ︸

F

(8.15)

のように整理できる．一般化力 Fα, Fβ の具体形は 9節で計算する．

8.4 運動方程式の 1階化

運動方程式は，加速度を含むので，数学的には 2階の常微分方程式となる．ところが，
コンピュータは 2階微分の処理が苦手なので，運動方程式を 1階の常微分方程式に見せか
けるテクニックが必要になる．これを運動方程式の 1階化という．
簡単のため，単振り子の運動方程式，

ml2ẍ = −mgl sinx (*)

を例にとる (xは角度)．この方程式の 2階微分 ẍを見かけ上消すため，1 階微分を他の変
数に置いた ẋ = yを連立する． ẋ = y

ml2ẍ = −mgl sinx

これに，変換式 ẋ = yを微分したもの ẍ = ẏを代入すると，ẋ = y

ẏ = −g
l sinx ※整理した

(**)

となり，見かけ上，2階微分が消せる．このような式変形，

2階 1連立 (*)
ダミー変数
=⇒
y=ẋ

1階 2連立 (**)

を，1階化という．1階化の前 (*)と後 (**)で変数の数こそ違うが，方程式としては等価
であり，同じ解を持つ 2)．
さらに，3階微分を含む場合

...
x = f(ẍ, ẋ, x) でも，2階微分までを別の変数 ẋ = y, ẍ = z

に置けば， 
ẋ = y

ẏ = z

ż = f(z, y, x)

(8.16)

のように 1階化できる．同様の手順で変数を増やせば，一般に，

n階 1連立
ダミー変数
=⇒ 1階 n連立

という 1階化が可能である．変数が多いときは，通し番号の変数，

x1 = x, x2 = ẋ, x3 = ẍ, · · · , xn =
dn−1x

dtn−1

を使うのが便利である．

例題 3.3 次の運動方程式を 1階化せよ．α̈ = h1(α, α̇, β, β̇)

β̈ = h2(α, α̇, β, β̇)
(8.17)

新しい変数は，x1 = α, x2 = α̇, x3 = β, x4 = β̇ とせよ．

▶解答例 新しい変数と，その微分 ẋ1 = α̇, ẋ2 = α̈, ẋ3 = β̇, ẋ4 = β̈ を使うと，
ẋ1 = x2 ∵ ẋ1 = α̇ = x2

ẋ2 = h1(x1, x2, x3, x4)

ẋ3 = x4 ∵ ẋ3 = β̇ = x4

ẋ4 = h2(x1, x2, x3, x4)

(8.18)

という 1階化が得られる．

実習 3.1 Code 4を実行せよ．(8.15)において Fα = Fβ = 0としたときの自由運動がア
ニメーション表示される．なお，プログラム中の運動方程式は，(8.15)を加速度について

q̈ = A−1
(
b+F

) (
≡ h =

[
h1

h2

])
(8.19)

のように解き，これを例題 3.3 p22のように 1階化した形式で書かれている．

2)(*)から (**)が導いたが，(**)から (*)も導ける．ゆえに方程式 (*)と方程式 (**)は，数学的に等価で
あり，同じ解を持つ．

9 一般化力とその応用

9.1 一般化力F = [Fi]

力の単位はニュートン (N)だった．こうした単位を無視して定義される力ベクトルF =

[Fi]を，一般化力 (generalized force) という．
例えば，平面上の剛体に作用させられる外力は，力 F = [Fi] [N] とトルク T [N·m] で
ある．これらを，単位を気にせず並べてしまった，

F = [Fi] =

F1 [N]

F2 [N]

T [N·m]

 (9.1a)

を，一般化力という．

9.2 一般化力の作用

一般化力の作用を知るには，対応する一般化座標の運動を想像すればよい．簡単のため，
平面上の剛体を考える．その運動法則mẌ = F , Iθ̈ = T を成分で見ると，

mẌ1 = F1

mẌ2 = F2

Iθ̈ = T

(9.1b)

である．ここで F1の作用とは，対応するX1を増減させる作用である．同様に，F2, T は
それぞれX2, θを増減させる．一般化力も同じである．一般化力の各成分Fiは，対応する
一般化座標の成分 qiを増減させる．

例題 3.4 例題 3.1の一般化座標 q = (q1, q2)に対する一般化力を F = (F1,F2)とする．
(a)と (b)のそれぞれについて，一般化力の各成分 Fiが，2節リンクのどの自由度を増減
させるか説明せよ．(b)の F2を関節トルクという．

▶解答例 F1 の効果は (a), (b)共通で，慣性系と第 1リンクのなす角 q1 を増減させる．こ
れに対して，(a)のF2は第 2リンクの角度 q2だけを増減させ，q1には作用しない．(b)のF2

は第 1リンクと第 2リンクの相対角 q2 を増減させる．

例題 3.5 例題 3.4の説明を踏まえて，F2がリンクに与える作用が，(a)絶対角方式と (b)

相対角方式でどのように違うか，考察せよ．

▶解答例 図 2のように，(a)絶対角方式の F2 は，絶対角 q2 を直接変化させるトルクなの
で，その作用の対象は第 2リンクのみである．これに対して，(b)相対角方式の F2 は，相対
角 q2を押し開く (閉じる)ように働くので，第 1リンクと 2リンクに対して，同時に逆向きの
トルク F2,−F2 が作用する．

図 2に示した構造を，もう少し詳しく補足しておこう．(a)絶対角方式では，モーター
の外枠は 2個とも，外部構造 (慣性系など)に対して回転してはならない．となると上側の
モータの設置方法が課題になる．例えば，外部構造にモータの外枠を固定し，回転体 (軸)

の回転を回転伝達用ケーブル 3)で伝えるような設計もありうるが，工業的にはあまり見ら
れない構造だと思う．その一方で，(b)相対角方式では，外部構造に固定するのは，下側

3)自転車のブレーキケーブルの高級バージョンみたいなやつ．

(a) 絶対角方式 (b) 相対角方式

図 2: 2節リンクにおける一般化力の作用

のモータのみでよい．上側のモータは，外枠を第 1リンクに固定し，回転軸を第 2リンク
に固定すればよい (逆でもいいが)．(b)相対角方式のモーターがリンクに発揮するトルク
を，関節トルクという．この構造はロボットの関節に応用されている．
関節トルクは，内力である．なぜなら，関節トルクは隣接するリンクに同時に逆向きの
トルク (例えば，F2 と −F2)を発生するから，総和すればキャンセルする．したがって，
外力を受けずに浮遊する 4)ロボットの姿勢を，こうした関節トルクで変更しても，重心運
動の軌跡は元のまま維持されることになる 5)．

9.3 一般化力の座標変換

以上，一般化力 Fiの作用を定性的に理解した．では，定量的にはどうだろう．一般化
力はどんな値をとるのか．これは，次の算法で計算できる．

算法 3.1 直交座標系における力 F = [Fi]とその着力点 x = [xi]を考える．着力点の直
交座標 xと，他の任意の一般化座標 q = [qi]との座標変換が，

x = x(q) =

x1(q1, · · · , qm)
...

xn(q1, · · · , qm)

 (9.2)

で与えられるとき，q = [qi]に関する一般化力F = [Fi]は，

Fi =
n∑

k=1

∂xk
∂qi

Fk =

[
∂x1
∂qi

, · · · , ∂xn
∂qi

]F1

...

Fn

 =:

(
∂x

∂qi

)T

F (9.3)

で計算できる．表記短縮のため，さらに Fiをベクトルにまとめると，

F ≡

F1

...

Fm

 =


∂x1
∂q1

· · · ∂xn
∂q1

...
. . .

...
∂x1
∂qm

· · · ∂xn
∂qm


F1

...

Fn

 =:

(
∂x

∂q

)T

F (9.4)

という表現を得る．∂x/∂qは座標変換 x(q)のヤコビ行列である．

4)重力はあってよい．
5)ただし，空気抵抗がなければ．空気抵抗という外力は，ロボットの形状に依存するので，ロボットの姿勢

が変れば外力が変化し，ゆえに重心の軌跡も変化する．

驚くべきことに，算法 3.1の F と xは，実は，直交座標系における力とその着力点で
なくてもよい．すなわち，算法 3.1は，次のように読み替えることができる．

算法 3.2 1つめの一般化座標 x = [xi]に関する一般化力を F = [Fi]とし，2つめの一
般化座標 q = [qi]に関する一般化力をF = [Fi]とするとき，両者の関係は，(9.2), (9.3)

で計算できる．

算法 3.1～3.2の力の換算法は，物理的な直感に自信が持てないときに非常に頼りにな
る．計算すれば答えが出るからだ．

9.4 車輪型倒立ロボット

8.3節 p20の車輪型倒立ロボットを考える．

例題 3.6 (車輪型倒立ロボットの一般化力) 関節トルク T が自由度 α, βにおよぼす一般化力
FT ≡ (Fα,Fβ)

T を求めよ．

▶解答例 トルク T は自由度 θ ≡ α− βに働く一般化力なので，これを T = Fθ と表記する．
このとき算法 3.1～3.2より，

Fα =
∂θ

∂α
Fθ = 1 · T = T, Fβ =

∂θ

∂β
Fθ = −1 · T = −T

となる．ゆえにFT = (T,−T )T を得る 6)．

以上のFT を，運動方程式 (8.15) p21の一般化力F に，

F = FT (9.5)

として代入すれば，トルク T を受けた車輪型倒立ロボットの運動方程式が得られる．
この系を倒立状態で安定化するために，関節トルク T に，次のような PD制御入力 7)

を与えることにする．
T = K1α+K2α̇+K3β +K4β̇ (9.6)

これにより，原点 x = rα = 0に収束する位置制御と，倒立姿勢 β = 0に収束する倒立制
御が達成される．

実習 3.2 (車輪型倒立ロボットの倒立制御) (K1,K2,K3,K4) = (1, 0.5, 40, 4)のときのロボッ
トの動きを確認せよ．このときロボットは原点で倒立安定化する．

▶解答例 Code 4の 7行目：

FT = [0; 0];

を，

T = 1*a + 0.5*da + 40*b + 4*db; FT = [T; -T];

に変更して実行すればよい．

6)肩の T は転置の T．
7)機械力学 [1] 12.3節を復習せよ．

10 接触と摩擦１（バウンド）

実例を見てしまうのがはやい．

実習 3.3 Code 5は，質量mで長さが lの均一な剛体棒が，床面でバウンドするシミュ
レーションである．Code 5を実行せよ．

10.1 床面でバウンドする棒

Code 5のシミュレータは，次の力学モデルを解いている．

f1,f2は端点が受ける外力で，後に，床からの反力と摩擦力が代入される．gは重力加速
度である．xは重心の位置ベクトル，θは棒の傾斜角であり，

u1 = − l

2

[
cos θ

sin θ

]
, u2 = −u1 (10.1)

は，重心 (この棒では中点)から見た f1,f2の着力点の位置ベクトルである．棒が十分に
細いと仮定すると，重心まわりの慣性モーメントは，I = ml2/3となる．
単純な系なので，解析力学ではなく，剛体の運動法則に基づいて運動方程式を求めよう．
まず，重心で力とトルクの集約すると，この棒に作用する合力は，

F = f1 + f2 +mg (10.2)

合トルクは，(“∧”は符号付き面積 8))

T = u1 ∧ f1 + u2 ∧ f2 (10.3)

8)機械力学 [1] 3章を復習せよ．

となる．これらを，剛体の運動方程式 9)に代入すると，mẍ = F = f1 + f2 +mg (ニュートン方程式)

Jθ̈ = T = u1 ∧ f1 + u2 ∧ f2 (オイラー方程式)
(10.4)

となる．これを解いてアニメーション表示したのがCode 5である．
ただし現時点では，外力 f1,f2の具体形が不明である．以下，床からの反力と摩擦力を
数式表現し，f1,f2に代入する方法について述べる．
以下，外力 f iを，垂直反力Riと，摩擦力 F iの和として，

f i = F i +Ri =

[
Fi

0

]
+

[
0

Ri

]
(i = 1, 2) (10.5)

と表す．垂直抗力Riの具体形は 10.2節で，摩擦力 F iの具体形は 11節で与える．

10.2 垂直抗力の数理モデル — ペナルティー法

自由落下する棒が，床に着く前までは，端点に受ける外力はないので f1 = f2 = O で
ある．ところが，床と接触したとたん，棒は床から反力 f1,f2 ̸= Oを受ける．反力に弾
かれた棒が，再び宙に舞うと，また f1 = f2 = Oに戻る．こうした状況は，どのように
数式表現できるだろうか？
簡単のため，図 3の左に示すように，自由落下する 1質点が床で反発する問題を考える．
こうした反発現象のモデルとして，広く実用されているのが，図 3の右である．床をばね
とダンパーで表し，質点と床が接触する位置を y = 0としている．こうしたトランポリン
型のモデル化手法を，ペナルティー法という．

図 3: 垂直抗力のモデル (ペナルティー法)

このモデルでは，質点は床下までめり込む．めり込んでいる間だけ，ばねとダンパーに
よる反力Rが発生すると考える．式で書くと，

R = R(y, ẏ) =

0 (y > 0)

−Kpy − Cpẏ (y ≤ 0)
(10.6)

9)機械力学 [1] 7章のニュートン・オイラー方程式を復習せよ．

である．Kpは質点が床にめり込んだときのばね定数，Cpは同じく粘性係数である．Kp

で床の硬さを調整し，Cpで反発係数を調整する．また，プログラミングでは，(10.6)の場
合分けをステップ関数で，次のように書くテクニックも使われる 10)．

R = R(y, ẏ) = U(−y){−Kpy − Cpẏ} (10.7)

関数 U(X)を，単位ステップ関数 (unit step function)と呼ぶ．定義は，

U(X) =

1 (X ≥ 0)

0 (X < 0)
(10.8)

である．式 (10.7)のなかでは，単位ステップ関数を左右反転して使うために，U(−y)とし
ている．
以上のモデルでは，Cpの粘性抵抗が，田んぼに足を突っ込んだときのように，足を引
き抜くときにも効く．これをトランポリンのように，引き抜く方向ではゼロと仮定して，

R = R(y, ẏ) = U(−y){−Kpy − U(−ẏ)Cpẏ} (10.9)

とすることもできる 11)．

10.3 シグモイド関数の活用

その他，プログラミング上のテクニックとして，ステップ関数の代用に，ステップ関数
の角を丸めたシグモイド関数：

Us(X) :=
1

1 + exp(−sX)
(10.10)

0

0.2

0.4

0.6

0.8

1

-0.4 -0.2 0 0.2 0.4

U
s
(
x
)

x

s=10

s=50

s=1000

を使う場合がある．sはステップの丸みを表わすパラメータであり，sを大きくすると丸
みが鋭くなる (s → ∞の極限でステップ関数となる)．この代用の理由だが，数値積分の
アルゴリズムはステップ関数のような不連続関数が苦手であり，それが原因で数値積分に
失敗することがある (滅茶苦茶な解がでる or エラーがでて解がでない)．このようなとき，
シグモイド関数を使って，エラーが解消する程度まで，徐々にステップの角を丸めていく．
もちろん，丸めすぎると，硬い床のモデルとはいえなくなるので，要注意．
10)Code 5 p32の「方法 1」
11)Code 5 p32の「方法 2」

10.4 床面で反発する棒の例題

式 (10.7)，(10.9)，(10.10)より，式 (10.5)における垂直抗力Riの成分Riは，

Ri = R(yi, ẏi) = Us(−yi){−Kpyi − Cpẏi} (i = 1, 2) (10.11)

または，

Ri = R(yi, ẏi) = Us(−yi){−Kpyi − U(−ẏi)Cpẏi} (i = 1, 2) (10.12)

となる．ただし，yiは，端点の位置ベクトル xi = x+uiの y 成分，ẏiは同じく速度ベク
トル，

ẋi = ẋ+ u̇i (i = 1, 2), u̇1 =
lθ̇

2

[
sin θ

− cos θ

]
, u̇2 = −u̇1 (10.13)

の y成分である．

11 接触と摩擦２（スリップ）

11.1 摩擦の数理モデル

物体を力Rで床に押し付けたときに発生する摩擦力は，

F = F (ẋ) =


−µR (ẋ > 0)

0 (ẋ = 0)

µR (ẋ < 0)

(11.1)

と書ける．ẋは床から見た物体の相対速度である．また，µを物体と床の動摩擦係数 (co-

efficient of dynamic friction)という．符号関数

sgn(X) :=


1 (X > 0)

0 (X = 0)

−1 (X < 0)

(11.2)

を使うと，(11.1)と同じことは，

F = F (ẋ) = −µR sgnẋ (11.3)

とも書ける．

11.2 シグモイド関数の活用

(11.1)または (11.3)は不連続関数なので，垂直抗力のときと同様に，数値計算上の問題
を引き起す．同様の対処法として，シグモイド関数 (10.10)を使って，(11.3)の不連続部
分を丸めるには，

F = F (ẋ) = −µR sgns(ẋ), sgns(ẋ) := 2Us(ẋ)− 1 (11.4)

のようにすればよい．0から 1まで変化するUsの高さを 2倍して縦に−1平行移動するこ
とで，−1から 0を経て 1に至る関数 sgns(ẋ)を作り出している．

11.3 床面で反発する棒の例題

式 (10.10)，(11.3)，(11.4) より，式 (10.5)における摩擦力 F iの成分 Fiは，

Fi = F (ẋi, yi, ẏi) = −µR(yi, ẏi) sgns(ẋi) (11.5)

となる．ただし，R(yi, ẏi)は，(10.11)，(10.12)の垂直抗力である．また，ẋiは端点の速
度ベクトル ẋi = ẋ+ u̇iの x成分である．
以上，式 (10.11)，(10.12)，(11.5)をまとめると，床との接触によって棒の端点が受け
る外力 f iの数式表現が，次のように得られる．

f i =

[
Fi

Ri

]
=

[
−µRi sgns(ẋi)

Us(−yi){−Kpyi − Cpẏi}

]
(i = 1, 2) (11.6)

または =

[
Fi

Ri

]
=

[
−µRi sgns(ẋi)

Us(−yi){−Kpyi − U(−ẏi)Cpẏi}

]
(i = 1, 2) (11.7)

11.4 応用 — 転倒する車輪型倒立ロボットの作成

11.4.1 転倒に必要なカラクリ

フィードバック制御で安定化された倒立ロボットが転倒するには，ある限界で制御を諦め
るカラクリが必要である．その一例として，振り子の傾斜角の絶対値 |β|が基準値βmax > 0

を超えたときに制御が打ち切られるような，フィードバック制御入力，

T ≡

K1α+K2α̇+K3β +K4β̇ (|β| ≤ βmax)

0 (それ以外)
(11.8)

を考えることができる．
垂直抗力 (10.7) p28のときと同様に，場合分けを関数で書くには，次のような台形関数

(trapezoidal function)を使えばよい．台形関数とは，両側で 0，中央付近で 1に立ち上が
る台形状の関数のことである．これは，左右対称な 2個のステップ関数 (10.8) p28を，左
右に平行移動してから掛け算することで，次のように構成できる．

trap(x;x0, w) ≡ U
(
− (x− x0) + w

)
· U
(
(x− x0) + w

)
(11.9)

x0は台形の中心，wは台形の幅の 1/2である．これを使うと，(11.8)と同じことは，

T ≡ trap(β; 0, βmax) · (K1α+K2α̇+K3β +K4β̇) (11.10)

と書ける．これで数値積分が不安定になる場合は，ステップ関数 U(x)の代りに，これを
丸めた Us(x)を使えばよい．

実習 3.4 Code 6を実行し，(11.10)を制御入力とする台車型倒立ロボットの運動を観察
せよ．指定された角度 b0で制御が打ち切られ，自由運動に移行してスイングを続ける様
子が見てとれる．

11.4.2 床面の設定

自由運動でスイングする振り子の先端を，ペナルティー法 p27で設けた床面にぶち当て
れば，床面に倒れこむ動作がシミュレートできる．床面との衝突の際，振り子の先端は力，

f =

[
F

R

]
(11.11)

を受ける．Rは床からの垂直抗力，F は摩擦力である．
これを運動方程式に組み込むための枠組みとして，振り子の先端に力 f = (fx, fy)

T を
作用させた次のモデルを導入する．簡単のため，棒や先端の太さは無視できるとしよう．

以上の f が角度α, βに及ぼす一般化力Ff ≡ (fα, fβ)
T を求めて，運動方程式 (8.15) p21

の Fα,Fβ に加算すれば，このモデルの運動方程式が得られることになる．
まず，一般化座標 q = (α, β)T から，振り子先端の直交座標 xmへの座標変換は，

xm = xm(α, β) =

[
xm(α, β)

ym(α, β)

]
=

[
rα+ l sinβ

r + l cosβ

]
(8.4) p20

であった．ゆえに，算法 3.1～3.2から変換公式：

Ff =

[
fα

fβ

]
=

[
∂xm
∂α

∂ym
∂α

∂xm
∂β

∂ym
∂β

][
fx

fy

]
=

[
r 0

l cosβ −l sinβ

]
f (11.12)

を得る．最後に，右辺の f をペナルティー法 (11.6) p30の要領で，

f =

[
fx

fy

]
=

[
F

R

]
=

[
−µR sgns(ẋm)

Us(−y){−Kpym − Cpẏm}

]
(11.13)

のように与えれば，振り子の先端と床との衝突がシミュレートされる．ただし

ẋm =

[
ẋm

ẏm

]
=

[
rα̇+ lβ̇ cosβ

−lβ̇ sinβ

]
(8.5) p20

である．
得られた床反力の効果Ff を加算して，運動方程式 (8.15) p21の一般化力F を

F = FT︸︷︷︸
倒立制御

+ Ff︸︷︷︸
床反力

(11.14)

とすれば，倒立ロボットの先端が床面に衝突する現象をシミュレートできる．
実習 3.5 Code 7を実行し，床の反力 (垂直抗力と摩擦力)を受ける台車型倒立ロボット
の運動を観察せよ．指定された角度 bmaxで制御が打ち切られて振り子の先端が床に落ち，
バウンドして停止する 12)．

12)Kp, Cp の調整が不十分で床にめり込みすぎかも．

A プログラム例

Code 4: “wip.sce” (Scilab)

1 clear;clf ();
2 ///// 運 動 方 程 式 の 定 義 /////
3 M=1, J=0.1, m=5, r=0.2, l=1, g=9.8;
4 function dx = eom(t,x)
5 a=x(1); da=x(2); b=x(3); db=x(4);
6 // T = 1*a + 0.5*da + 40*b + 4*db; FT = [T; -T];
7 FT = [0; 0];
8 A = [(M+m)*r^2+J, m*l*r*cos(b); ...
9 m*l*r*cos(b), m*l^2];

10 bb = [m*l*r*db^2 *sin(b); m*g*l*sin(b)] + FT;
11 h = A\bb;
12 dx(1) = x(2); dx(2) = h(1);
13 dx(3) = x(4); dx(4) = h(2);
14 endfunction
15 ///// 運 動 方 程 式 を 解 く /////
16 tn =1001; tt=linspace (0 ,0.01*tn ,tn);
17 x0=[0; 0; 0.2; 0];
18 xx=ode( x0 , 0, tt , eom );
19 ///// ロ ボ ッ ト 描 画 用 の 関 数 /////
20 function R=Rot(a) //回 転
21 R=[cos(a),-sin(a);...
22 sin(a), cos(a)];
23 endfunction
24 function yy=Trans(xx,rr) //平 行 移 動
25 yy(1 ,:)=xx(1,:) + rr(1)* ones(xx(1 ,:));
26 yy(2 ,:)=xx(2,:) + rr(2)* ones(xx(2 ,:));
27 endfunction
28 function draw_robot(a,b) //ロ ボ ッ ト 描 画
29 x=r*a; //車 輪 中 心 の 水 平 変 位
30 sn=5; sq=linspace (0 ,2*%pi*(sn -1)/sn,sn);
31 spoke=r*[cos(sq);sin(sq)]; //原 点 に あ る ス ポ ー ク の 外 周 点
32 spoke=Rot(-a) * spoke; //角 度 a回転したスポークの外周点
33 spoke=Trans(spoke ,[x;r]); //位 置 xのスポークの外周点
34 xv=[x*ones (1:sn);spoke (1 ,:)]; yv=[r*ones (1:sn);spoke (2 ,:)];
35 xsegs(xv,yv ,2); xarc(x-r, 2*r, 2*r, 2*r, 0, 360*64); //ス ポ ーク; 車 輪
36 rod=[0,0;0,l]; //原 点 に あ る 棒 の 両 端 点
37 rod=Rot(-b) * rod; //角 度 β だ け 回 転 し た 棒 の 両 端 点
38 rod=Trans(rod ,[x;r]); //位 置 xの棒の両端点
39 plot( rod(1,:), rod(2,:),"r-"); //棒 の 描 画
40 p=gce(); p.children.thickness =2;//直 前 の 描画; 線 の 太 さ
41 g=gca(); g.data_bounds =[-4,-1;4,2]; //座 標 軸 の 範 囲
42 g.isoview ="on"; xgrid (4); //縦 横比1; グ リ ッド;
43 endfunction
44 a=x0(1); b=x0(3); drawlater; draw_robot(a,b); drawnow;
45 sleep (2000); //2 s待ち
46 realtimeinit (0.01); //ア ニ メ ー シ ョ ン の 時 間 刻 み
47 for i=1:10: tn
48 realtime(i);
49 drawlater; clf (); //描 画 延期; 描 画 消去;
50 a=xx(1,i); b=xx(3,i);
51 draw_robot(a,b);
52 xlabel(sprintf ("%d / %d", i, tn));
53 drawnow; //画 面 更新;
54 end

Code 5: “stick.sce” (Scilab)

1 clear;clf ();
2 ///// 運 動 方 程 式 /////
3 m=1; l=1; J=m*l^2/3; g = 9.8; mu=0.3;
4 function y = torq(x,f)
5 y = det([x,f]);
6 endfunction
7 function y = signum( x )
8 y = (2.0/(1+ exp(-1e4*x)) -1.0); //符 号 関 数
9 endfunction

32



10 function y = step( x )
11 y = 1.0/(1.0+ exp(-1e4*x)); //ス テ ッ プ 関 数
12 endfunction
13 function fy = R( y, dy ) //床 反 力
14 Kp=8e3; Cp=25; fy = step(-y)*(-Kp*y -Cp*dy); //『 方法1』
15 //Kp=12e3; Cp=80; fy = step(-y)*(-Kp*y -step(-dy)*Cp*dy); //『 方法2』
16 endfunction
17 function fx = F( dx, R ) //摩 擦 力
18 fx = -signum(dx)*mu*R;
19 endfunction
20 function dq = eom(t,q)
21 x=q(1); dx=q(2); y=q(3); dy=q(4); a=q(5); da=q(6);
22 u1 = -(l/2)*[ cos(a); sin(a)]; u2 = -u1;
23 du1 =-(l/2)*da*[-sin(a); cos(a)]; du2 = -du1;
24 X1 = [x;y]+u1; //端点1の 位 置
25 X2 = [x;y]+u2; //端点2の 位 置
26 dX1 = [dx;dy]+du1; //端点1の 速 度
27 dX2 = [dx;dy]+du2; //端点2の 速 度
28 R1 = R(X1(2), dX1 (2)); //床 垂 直 抗力1
29 R2 = R(X2(2), dX2 (2)); //床 垂 直 抗力2
30 F1 = F(dX1(1), R1); //床 摩 擦力1
31 F2 = F(dX2(1), R2); //床 摩 擦力2
32 f1 = [F1;R1]; f2 = [F2;R2]; //端 点 に 作 用 す る 力
33 F = f1 + f2 + m*[0;-g]; //合 力
34 T = torq(u1 , f1) + torq(u2 , f2); //合 ト ル ク
35 dq(1) = dx; dq(2) = F(1)/m;
36 dq(3) = dy; dq(4) = F(2)/m;
37 dq(5) = da; dq(6) = T/J;
38 endfunction
39 ///// 運 動 方 程 式 を 解 く /////
40 n=130; tt=linspace (0 ,0.05*(n-1),n);
41 x0 = [0; 0; 5; 0; %pi/4; 0];
42 xx=ode( x0 , 0, tt , eom );
43 ///// ア ニ メ ー シ ョ ン す る /////
44 function draw_mech(x,y,th)
45 g=gca(); //座 標 軸 の 取 得
46 g.data_bounds (: ,1)=[ -2;4]; // x軸の範囲
47 g.data_bounds (: ,2)=[0;5]; // y軸の範囲
48 g.isoview ="on"; //縦 横比1; グ リ ッド;
49 xset(" color ",1); //基 本 色 黒
50 xsegs ([ -3;5] ,[0;0] ,9); // x軸
51 xsegs ([0;0] ,[ -0.5;5.5] ,9); // y軸
52 u1 = -(l/2)*[ cos(th); sin(th)]; u2 = -u1;
53 X1 = [x;y]+u1; X2 = [x;y]+u2;
54 xsegs ([X1(1);X2(1)],[X1(2);X2(2)] ,5); //ス テ ィ ッ ク
55 p=gce(); p.thickness =2; //直 前 の 描画; 線 の 太 さ
56 endfunction
57 x=x0(1); y=x0(3); th=x0(5);
58 draw_mech(x,y,th); drawnow; //描 画 実行; 画 面 更新;
59 // xclick (); //マ ウ ス ク リ ッ ク 待 ち
60 sleep (2000);
61 realtimeinit (0.05); //ア ニ メ ー シ ョ ン の 時 間 刻 み
62 for i=1:n
63 realtime(i);
64 drawlater (); clf (); //描 画 延期; 描 画 消去;
65 x=xx(1,i); y=xx(3,i); th=xx(5,i);
66 draw_mech(x,y,th); //描 画 消去; 棒 を 描く;
67 drawnow; //画 面 更 新
68 end
69 // xs2eps (0," stick.eps ");

Code 6: “wip-fall-free.sce” (Scilab)

1 clear;clf ();
2 /////////////// 諸 元 ///////////////
3 M=1, J=0.1, m=5, r=0.2, l=1, g=9.8;
4 bmax =0.29; //転 倒 角
5 ////////////////////////////////////
6 ///// ス テ ッ プ 関 数 ほ か /////
7 function y = step( x )
8 s = 100;
9 y = 1.0./(1.0+ exp(-s*x)); //ス テ ッ プ 関 数

10 endfunction
11 function y = trap( x, x0, w )
12 y = step((x-x0)+w).* step(-(x-x0)+w); //台 形 関 数
13 endfunction
14 ///// 運 動 方 程 式 の 定 義 /////
15 function dx = eom(t,x)
16 a=x(1); da=x(2); b=x(3); db=x(4);
17 T = trap(b,0,bmax )*(1*a + 0.5*da + 40*b + 4*db); //制 御 の 打 ち 切 り

18 FT = [T; -T]; //ト ル ク → 一 般 化 力
19 A = [(M+m)*r^2+J, m*l*r*cos(b); ...
20 m*l*r*cos(b), m*l^2];
21 bb = [m*l*r*db^2 *sin(b); m*g*l*sin(b)] + FT;
22 h = A\bb; //加 速 度 の 右 辺
23 dx(1) = x(2); dx(2) = h(1);
24 dx(3) = x(4); dx(4) = h(2);
25 endfunction
26 ///// 運 動 方 程 式 を 解 く /////
27 tn =1001; tt=linspace (0 ,0.01*tn ,tn);
28 x0=[0; 0; 0; 1];
29 xx=ode( x0 , 0, tt , eom );
30 ///// ロ ボ ッ ト 描 画 用 の 関 数 /////
31 function R=Rot(a) //回 転
32 R=[cos(a),-sin(a);...
33 sin(a), cos(a)];
34 endfunction
35 function yy=Trans(xx,rr) //平 行 移 動
36 yy(1,:)=xx(1,:) + rr(1)* ones(xx(1 ,:));
37 yy(2,:)=xx(2,:) + rr(2)* ones(xx(2 ,:));
38 endfunction
39 function draw_robot(a,b) //ロ ボ ッ ト 描 画
40 x=r*a; //車 輪 中 心 の 水 平 変 位
41 sn=5; sq=linspace (0 ,2*%pi*(sn -1)/sn,sn);
42 spoke=r*[cos(sq);sin(sq)]; //原 点 に あ る ス ポ ー ク の 外 周 点
43 spoke=Rot(-a) * spoke; //角 度 a回転したスポークの外周点
44 spoke=Trans(spoke ,[x;r]); //位 置 xのスポークの外周点
45 xv=[x*ones (1:sn);spoke (1 ,:)]; yv=[r*ones (1:sn);spoke (2 ,:)];
46 xsegs(xv,yv ,2); xarc(x-r, 2*r, 2*r, 2*r, 0, 360*64); //ス ポ ーク; 車 輪
47 rod=[0,0;0,l]; //原 点 に あ る 棒 の 両 端 点
48 rod=Rot(-b) * rod; //角 度 β だ け 回 転 し た 棒 の 両 端 点
49 rod=Trans(rod ,[x;r]); //位 置 xの棒の両端点
50 plot( rod(1,:), rod(2,:),"r-"); //棒 の 描 画
51 p=gce(); p.children.thickness =2; //直 前 の 描画; 線 の 太 さ
52 g=gca(); g.data_bounds =[-4,-1;4,2]; //座 標 軸 の 範 囲
53 g.isoview ="on"; xgrid (4); //縦 横比1; グ リ ッド;
54 endfunction
55 a=x0(1); b=x0(3); drawlater; draw_robot(a,b); drawnow;
56 sleep (2000); //2 s待ち
57 realtimeinit (0.01); //ア ニ メ ー シ ョ ン の 時 間 刻 み
58 for i=1:10: tn
59 realtime(i);
60 drawlater; clf (); //描 画 延期; 描 画 消去;
61 a=xx(1,i); b=xx(3,i);
62 draw_robot(a,b);
63 xlabel(sprintf ("%d / %d", i, tn));
64 drawnow; //画 面 更新;
65 end

Code 7: “wip-fall-floor.sce” (Scilab)

1 clear;clf ();
2 /////////////// 諸 元 ///////////////
3 M=1, J=0.1, m=5, r=0.2, l=1, g=9.8;
4 bmax =0.29; //転 倒 角
5 mu=0.3; //摩 擦 係 数
6 ////////////////////////////////////
7 ///// ス テ ッ プ 関 数 ほ か /////
8 function y = step( x )
9 s = 100;

10 y = 1.0./(1.0+ exp(-s*x)); //ス テ ッ プ 関 数
11 endfunction
12 function y = trap( x, x0, w )
13 y = step((x-x0)+w).* step(-(x-x0)+w); //台 形 関 数
14 endfunction
15 function y = sgn( x )
16 s = 1e4;
17 y = (2.0/(1+ exp(-s*x)) -1.0); //符 号 関 数
18 endfunction
19 ///// 床 か ら の 反 力 （ ペ ナ ル テ ィ ー 法 ） /////
20 function ff = yuka(x)
21 global r l mu;
22 a=x(1); da=x(2); b=x(3); db=x(4);
23 xm = [r*a+l*sin(b); r+l*cos(b)];
24 dxm = [r*da+l*db*cos(b); -l*db*sin(b)];
25 Kp=1e4; Cp=90; //こ れ ら の 数 値 を 変 え る と 床 の 性 質 が 変 わ る
26 R = step(-xm(2))*( -Kp*xm(2) -Cp*dxm (2)); //垂 直 抗 力
27 F = -mu*R*sgn(dxm (1)); //摩 擦 力
28 ff = [F; R]; //床 か ら の 反 力
29 endfunction
30 ///// 運 動 方 程 式 の 定 義 /////

31 function dx = eom(t,x)
32 a=x(1); da=x(2); b=x(3); db=x(4);
33 /// 制 御 の 打 ち 切 り×(位 置 制 御 + 倒 立 制御) ///
34 T = trap(b,0,bmax )*(1*a + 0.5*da + 40*b + 4*db);
35 FT = [T; -T]; //ト ル ク → 一 般 化 力
36 /// ペ ナ ル テ ィ ー 法 に よ る 床 反 力 ///
37 ff=yuka([a; da; b; db]);
38 Ff = [r, 0; l*cos(b), -l*sin(b)]*ff; //床 反 力 → 一 般 化 力
39 A = [(M+m)*r^2+J, m*l*r*cos(b); ...
40 m*l*r*cos(b), m*l^2];
41 bb = [m*l*r*db^2 *sin(b); m*g*l*sin(b)] + FT + Ff;
42 h = A\bb; //加 速 度 の 右 辺
43 dx(1) = x(2); dx(2) = h(1);
44 dx(3) = x(4); dx(4) = h(2);
45 endfunction
46 ///// 運 動 方 程 式 を 解 く /////
47 tn=401; tt=linspace (0 ,0.01*tn ,tn);
48 x0=[0; 0; 0; 1];
49 xx=ode( x0 , 0, tt , eom );
50 ///// ロ ボ ッ ト 描 画 用 の 関 数 /////
51 function R=Rot(a) //回 転
52 R=[cos(a),-sin(a);...
53 sin(a), cos(a)];
54 endfunction
55 function yy=Trans(xx,rr) //平 行 移 動
56 yy(1 ,:)=xx(1,:) + rr(1)* ones(xx(1 ,:));
57 yy(2 ,:)=xx(2,:) + rr(2)* ones(xx(2 ,:));
58 endfunction
59 function draw_robot(a,b) //ロ ボ ッ ト 描 画
60 x=r*a; //車 輪 中 心 の 水 平 変 位
61 sn=5; sq=linspace (0 ,2*%pi*(sn -1)/sn,sn);
62 spoke=r*[cos(sq);sin(sq)]; //原 点 に あ る ス ポ ー ク の 外 周 点
63 spoke=Rot(-a) * spoke; //角 度 a回転したスポークの外周点
64 spoke=Trans(spoke ,[x;r]); //位 置 xのスポークの外周点
65 xv=[x*ones (1:sn);spoke (1 ,:)]; yv=[r*ones (1:sn);spoke (2 ,:)];
66 xsegs(xv,yv ,2); xarc(x-r, 2*r, 2*r, 2*r, 0, 360*64); //ス ポ ーク; 車 輪
67 rod=[0,0;0,l]; //原 点 に あ る 棒 の 両 端 点
68 rod=Rot(-b) * rod; //角 度 β だ け 回 転 し た 棒 の 両 端 点
69 rod=Trans(rod ,[x;r]); //位 置 xの棒の両端点
70 plot( rod(1,:), rod(2,:),"r-"); //棒 の 描 画
71 p=gce(); p.children.thickness =2; //直 前 の 描画; 線 の 太 さ
72 g=gca(); g.data_bounds =[-4,-1;4,2]; //座 標 軸 の 範 囲
73 g.isoview ="on"; xgrid (4); //縦 横比1; グ リ ッド;
74 endfunction
75 a=x0(1); b=x0(3); drawlater; draw_robot(a,b); drawnow;
76 sleep (2000); //2 s待ち
77 realtimeinit (0.01); //ア ニ メ ー シ ョ ン の 時 間 刻 み
78 for i=1:10: tn
79 realtime(i);
80 drawlater; clf (); //描 画 延期; 画 面 消去;
81 a=xx(1,i); b=xx(3,i);
82 draw_robot(a,b);
83 xlabel(sprintf ("%d / %d", i, tn));
84 drawnow; //画 面 更新;
85 end

